

# **ADVPL III – Rotinas Web**

**Agosto/2010**

## Conteúdo

Capítulo 01 – Introdução .....	3
<b>Objetivos do Treinamento</b> .....	3
<b>Metas do Treinamento</b> .....	3
<b>Conteúdo</b> .....	3
Capítulo 02 – Manipulação de E-mail.....	4
<b>Parâmetros de Email</b> .....	4
<b>Passos para envio de email</b> .....	5
<b>Exercícios email</b> .....	6
Capítulo 03 – Manipulação de FTP .....	7
<b>Funções de FTP</b> .....	7
Capítulo 04 – Workflow .....	11
<b>Configuração</b> .....	11
<b>Criação do HTML</b> .....	17
<b>Criação de Processos em ADVPL</b> .....	18
<b>Classe TWFPPROCESS</b> .....	18
Capítulo 05 – WEB Service.....	25
<b>Configuração</b> .....	26
<b>Exercício</b> .....	34
Capítulo 06 – ADVPL ASP.....	35
<b>Características do ADVPL ASP – Arquivos .APH</b> .....	35
<b>Desenvolvimento de Funções .APL</b> .....	29

# Capítulo 01 – Introdução

- Objetivos do treinamento;
- Metas do treinamento;
- Conteúdo;

## Objetivos do Treinamento

Conhecer e empregar adequada e eficazmente os conceitos e funcionalidades do Protheus – ROTINAS DA WEB.

## Metas do Treinamento

Habilitar o aluno do curso de ADVPL III – Rotinas da WEB, a obter o conhecimento sobre a linguagem AdvPL – Advanced Protheus Language, voltada para a WEB e utilizada na programação do Protheus da Microsiga, um dos softwares de gestão mais utilizados em toda a América Latina. O conhecimento do uso desta linguagem permite a elaboração de customizações e o desenvolvimento de novas soluções.

## Conteúdo

- Email
- FTP
- Workflow
- Web Service
- ADVPLASP

## Capítulo 02 – Manipulação de E-mail

A linguagem ADVPL permite o envio e o recebimento de mensagens através de e-mails. Veremos a seguir as funções para manipulação de e-mails e parâmetros configurados previamente que podemos também utilização no envio de emails

### Parâmetros de Email

Parâmetro	Descrição	conteúdo
MV_RELACNT	Conta a ser utilizada no envio de E-Mail para os relatórios	workflow@advpl.com.br
MV_RELAPSW	Senha para autenticação no servidor de e-mail	escola1010
MV_RELAUTH	Determina se o Servidor necessita de Autenticação.	.T.
MV_RELFROM	E-mail utilizado no campo FROM no envio de relatórios por e-mail	workflow@advpl.com.br
MV_RELPSW	Senha da Conta de E-Mail para envio de relatórios	escola1010
MV_RELSERV	Nome do Servidor de Envio de E-mail utilizado nos relatórios	smtp.advpl.com.br
MV_WFSMTP	Nome do Servidor SMTP(envio de mensagem	smtp.advpl.com.br
MV_WFACC	Conta de Acesso(ex:informática)	workflow@advpl.com.br
MV_WFPASS W	Senha da Conta do Workflow	escola1010

Exemplo de utilização:

```
Private cSrv := GetMV("MV_RELSERV")
Private cMail := GetMV("MV_RELACNT")
Private cPass := GetMV("MV_RELPSW")
Private lAuth := GetMv("MV_RELAUTH")
Private cMailA := GetMV("MV_RELACNT")
Private cPswA := GetMV("MV_RELAPSW")
```

Ou

```
Private cSrv := GetMV("MV_WFSMTP")
Private cMail := GetMV("MV_WFACC")
Private cPass := GetMV("MV_WFPASSW")
Private lAuth := GetMv("MV_RELAUTH")
Private cMailA := GetMV("MV_RELACNT")
Private cPswA := GetMV("MV_RELAPSW")
```

## Passos para envio de email

- Conectar ao Servidor SMTP  
CONNECT SMTP SERVER cSrv ACCOUNT cMail PASSWORD cPass RESULT IResul
- Verificar se necessita de autenticação  
lOk := MailAuth(cEmail,cPass)
- Verificar mensagens da conexão  
GET MAIL ERROR cError
- Enviar o email  
SEND MAIL FROM **cDe** TO **cPara** SUBJECT **cAssunto** BODY **cMsg** CC **cCC** CCO **cCCO**;  
ATTACHMENT **cAnexo** RESULT **ISend**
- Desconectar do Servidor  
DISCONNECT SMTP SERVER

### Exemplo de um programa com Email

```
# include 'protheus.ch'
cServer      := GETMV("MV_WFSMTP")
cAccount     := GETMV("MV_WFACC")
cPassword    := GETMV("MV_WFPASSW")
cFrom        := GETMV("MV_RELFROM")
cDest        := "cris@advpl.com.br"

CONNECT SMTP SERVER cServer ACCOUNT cAccount PASSWORD cPassword Result lConectou
lAuth        := GetMv("MV_RELAUTH")
If lAuth
    lOk := MailAuth( cAccount ,GetMV("MV_RELAPSW"))
    If !lOk
        lOk := QAGetMail()
    EndIf
EndIf
cAssunto := "Teste de corpo de email"
SEND MAIL FROM cFrom TO cDest SUBJECT "Teste Assunto";
        BODY cAssunto ATTACHMENT "\system\error.log" RESULT lEnviado

If !lEnviado
    GET MAIL ERROR _cMensagem
    Alert(_cMensagem)
Endif

DISCONNECT SMTP SERVER Result lDisConectou
```

## Exercícios email

1. Fazer um programa que envie para você mesmo um email para cada cliente com as seguintes informações:

**Código:** 999999

**Nome:** XXXXXXXXXXXXXXXX

**Endereço:** XXXXXXXXXXXXXXXX

**Telefone:** 99 9999 9999

No Assunto coloque o Nome reduzido do cliente

Ex. HTML <strong> Negrito </strong> <em> Italico </em>

2. Fazer um programa que envie para você mesmo um único email com todos os fornecedores com as seguintes informações:

<b>Código</b>	<b>Loja</b>	<b>Nome</b>	<b>Endereço</b>
999999	99	XXXXXXXXXX	XXXXXXXXXXXXXXXXXX
999999	99	XXXXXXXXXX	XXXXXXXXXXXXXXXXXX
999999	99	XXXXXXXXXX	XXXXXXXXXXXXXXXXXX

No Assunto coloque o Nome reduzido do fornecedor

Ex. HTML

```
<table>
<tr><td>Coluna1 </td><td>Coluna2 </td><td>Coluna3 </td><td>Coluna4 </td></tr>
<tr> <td>lin1col1</td><td>lin1col2</td><td>lin1col3 </td><td>lin1col4</td></tr>
<tr> <td>lin2col1</td><td>lin2col2</td><td>lin2col3 </td><td>lin2col4</td></tr>
</Table>
```

## Capítulo 03 – Manipulação de FTP

A linguagem ADVPL permite o download e upload de arquivos através de FTP e também manipulação de arquivos no FTP, como alteração, exclusão, renomear arquivos e trocar de diretórios dentro do FTP.

### Funções de FTP

- **FTPConnect** - Conecta a um FTP

Sintaxe:

lRet := FTPConnect ( cServer, nPorta, cUser, cPass )

Argumento	Obrig	Tipo	Descrição
cServer	Sim	C	Servidor de FTP
nPorta	Sim	N	Porta de FTP, Default 21
cUser	Sim	C	Nome de usuário de FTP
cPass	Sim	C	Senha de FTP
lRet		L	.T. ou .F.

**Exemplo:**

lRet := FTPConnect("ftp.advpl.com.br", 21, "aluno", "123456")

- **FTPDiretory** - Retorna conteúdo do diretório corrente do FTP

Sintaxe:

aRet := FTPDiretory ( cMascara , cAttr )

Argumento	Obrig	Tipo	Descrição
cMascara	Sim	C	Máscara dos arquivos a retornar, ex. "*.*)"
cAttr	Não	C	Atributo dos arquivos a retornar
aRet		A	array com arquivos que obedecem os argumentos

**Exemplo:**

aRet := FTPDiretory("\*.\*)", "R")

- **FTPUpload** - Sobe arquivos para o FTP

Sintaxe:

IRet := FTPUpload( cFileLocal, cFileFTP )

Argumento	Obrig	Tipo	Descrição
cFileLocal	Sim	C	Caminho completo no server, o caminho deve estar dentro do rootpath.
cFileFTP	Sim	C	Nome do arquivo no FTP
IRet		A	T. ou .F.

**Exemplo:**

IRet := FTPUpload("\system\error.log", "error.log" )

- **FTPDownload** - Baixa arquivos do FTP

Sintaxe:

IRet := FTPDownLoad ( cFileLocal, cFileFTP )

Argumento	Obrig	Tipo	Descrição
cFileLocal	Sim	C	Caminho completo no server, o caminho deve estar dentro do rootpath.
cFileFTP	Sim	C	Nome do arquivo no FTP
IRet		A	T. ou .F.

**Exemplo:**

IRet := FTPDownLoad("\system\error.log", "error.log" )

- **FTPErase** - Apaga arquivo no FTP no diretório corrente

Sintaxe:

IRet := FTPErase( cFileFTP )

Argumento	Obrig	Tipo	Descrição
cFileFTP	Sim	C	Nome do arquivo no FTP que se quer apagar
IRet		A	T. ou .F.

**Exemplo:**

IRet := FTPErase("error.log")

- **FTPRenameFile** - Renomeia arquivo no FTP

Sintaxe:

IRet := FTPRenameFile(cFileOri, cFileNew)



Argumento	Obrig	Tipo	Descrição
cFileOri	Sim	C	Nome do Arquivo original
cFileNew	Sim	C	Nome novo para o arquivo
lRet		A	T. ou .F.

**Exemplo:**

lRet := FTPRenameFile("sigafin.xnu", "sigafin.old")

- **FTPSendFile** - Envia arquivo para o FTP

Sintaxe:

lRet := FTPSendFile( cDirFTP, cFileOri, cFileNew )

Argumento	Obrig	Tipo	Descrição
cDirFTP	Sim	C	Caminho escolhido
lRet		A	T. ou .F.

**Exemplo:**

lRet := : FTPDirChange("\tmp")

- **FTPGetCurDir** - Retorna o nome do diretório corrente do FTP

Sintaxe:

cRet := FTPGetCurDir( )

- **FTPDisconnect** - Desconecta o FTP, retorna .T. ou .F.

Sintaxe:

lRet := FTPDisconnect( )

**Exemplo de um programa com FTP**

```
# include 'protheus.ch'

User Function SendFtp()
Local cServer := 'ftp.totv.com.br'
Local nPorta := 21
Local cUser := 'totv'
Local cPass := 'escola1010'
Local aRetDir := {}

cpyt2s('c:\PRcris.csv', '\system\')

If FTPConnect( cServer, nPorta, cUser, cPass )
    Conout('FTP conectado!')
```

Continua...

Continuação...

```
Conout('Diretorio FTP',FTPGetCurDir())
Conout('FTPDirChange',FTPDirChange('aluno01'))
Conout('Diretorio FTP',FTPGetCurDir())

If FTPUpload( "\system\sigaфин.xnu" , "sigaфин.xnu")
    Conout('UpLoad Ok! ')
Else
    Conout('Falha UpLoad!')
EndIf
If FTPRenameFile("sigaфин.xnu", "sigaфин.old")
    Conout('Arquivo Renomeado! xnu para old')
Else
    Conout('Falha REnomear!')
EndIf
If FTPErase("sigaфин.old")
    Conout('Exclusão Ok! ' + "sigaфин.old")
Else
    Conout('Falha Exclusão!')
EndIf

aDirFTP := FTPDIRECTORY( "", )
aDirLoc := DIRECTORY( "\system\*.xnu", )

For I:=1 to 10 //Len(aDirLoc)
    If FTPUpload( "\system\" + aDirLoc[I][1], aDirLoc[I][1] )
        Conout('UpLoad Ok! ' + aDirLoc[I][1])
    Else
        Conout('Falha UpLoad!' + aDirLoc[I][1])
    EndIf
Next
For I:= 1 to LEN(aDirFTP)
    If FTPDownload( '\system\' + aDirFTP[I][1], aDirFTP[I][1] )
        Conout('Download Ok! ' + aDirFTP[I][1])
    Else
        Conout('Falha DownLoad ' + aDirFTP[I][1])
    EndIf
Next
FTPDisconnect()
Else
    Conout('Falha Conexao!')
EndIf

Return .T.
```

## Capítulo 04 – Workflow

É a automação de processos de negócio, utilizando-se o email como forma de comunicação nos processos críticos da organização, evitando gargalos e deixando o sistema mais ágil e seguro. Essa comunicação é feita a partir de um processo ativado por alguma rotina processada pelo usuário.

### Benefícios:

- Aumento da eficiência eliminando passos desnecessários.
  - Melhor gerenciamento dos processos.
  - Melhora nos serviços ao cliente.
  - Flexibilidade.
1. Forte rastreabilidade dos processos: pode-se saber a atual situação de qualquer processo através da consulta da rastreabilidade.
  2. Sistema Protheus e torna ativas, e a informação e a ação necessária a ser tomada é levada ao usuário e não o contrário.
  3. Controle de *time-out* dos processos.
  4. Qualquer processo existente pode ser automatizado através da construção de um Workflow.
- As ações a serem tomadas pelas pessoas em cobrança podem ser realizadas através de quatro meios, evitando que um processo tenha seu ciclo interrompido por falta de comunicação, são eles:
- e-mail: os processos são respondidos através de um cliente de e-mail homologado.
  - *browser* de internet: através do Internet Explorer.
  - Protheus: através do recurso de *Messenger* do Protheus.
  - Pontos de entrada: através de pontos de entradas existentes no sistema.

### Configuração

A configuração do Workflow é feita no módulo configurador (Ambiente - Workflow - Contas de Email), e em (Ambiente - Workflow - Parâmetros WF)

Os parâmetros são atualizados no cadastro de parâmetros (SX6)

Correio			
Opção	Item	Parâmetro	Descrição
Caixa de correio	Conta	MV_WFMLBO X	Caixa de correio a ser utilizada pelo workflow para o envio e recebimento de mensagens.
	Envia figura do html como anexo da mensagem	MV_WFIMAG E	Recurso ainda não disponível.
Composição da mensagem	Envia html no corpo da mensagem	MV_WFHTML	Selecionando esta opção, o html passara a fazer parte do corpo da mensagem. O contrário, irá como anexo.

	Envio automático	MV_WFSNDA U	Após a criação do processo, o workflow enviará a mensagem imediatamente.
	Usa Java Script	MV_WFJAVA S	No html, o Workflow adicionará rotinas semi-prontas em Java script.

Processos			
Opção	Item	Parâmetro	Descrição
Execução de retornos	Execução(ões) de retorno(s) simultâneos de processos.	MV_WFMAXJ B	Define a quantidade de processos de retorno que poderão ser executados por vez.
	Reativar processos automaticamente	MV_WFREAC T	Caso esta opção seja selecionada, ocorrerá erro na execução de retornos. O Workflow reativará o processo imediatamente para ser executado de novo. Caso contrário, será reativado somente quando o <b>Scheduler</b> for reiniciado.
Tratamento de erros	Usar TRANSAÇÃO na execução de funções de "RETORNO" e "TIMEOUTs"	MV_WFTRAN S	Habilita o recurso de transação com a finalidade de conservar a integridade dos dados em caso de falha de execução.

Notificação			
Opção	Item	Parâmetro	Descrição
E-mail do administrador	Endereço	MV_WFADMIN	Endereço eletrônico do administrador do sistema. Separe cada email por ponto e virgula (;) <i>Exemplo: user1@advpl.com.br; user2@advpl.com.br</i>
	Quando ocorrer erro ao executar funções "Retorno" e Timeout".	MV_WFNF001	Notificar por e-mail a lista de endereços dos administradores sobre o erro ocorrido.
Enviar notificação	Ao reativar processos pendentes.	MV_WFNF002	Notificar por e-mail a lista de endereços dos administradores no momento em que forem reativados os processos que ocorreram erro.
	Ao receber mensagens não reconhecidas.	MV_WFNF003	Notificar por e-mail a lista de endereços dos administradores sobre as mensagens não reconhecidas pelo Workflow.

<b>Messenger</b>			
<b>Opção</b>	<b>Item</b>	<b>Parâmetro</b>	<b>Descrição</b>
<b>Browser Internet</b>	Caminho	MV_WFBROWS	Arquivo executável do <i>browser</i> Internet que deverá estar no <i>path</i> da estação.
	Servidor	MV_WFBRWSR	IP ou Nomed PIPE do servidor Protheus para uso do serviço http. Adicione ":" + a porta, caso seja diferente do padrão.
<b>Diretório HTTP</b>	Caminho	MV_WFDHTTP	Diretório de trabalho do serviço http. Verifique o identificador " <b>Path=</b> " na seção " <b>[HTTP]</b> " do arquivo <b>mp8Srv.ini</b> para obter o diretório de trabalho.
<b>Habilitar</b>	Habilitar Messenger automaticamente . (próximo <i>logon</i> )	MV_WFMESSE	O <i>messenger</i> será executado automaticamente no próximo <i>login</i> de qualquer ambiente Protheus.

Após toda essa configuração você deve configurar o Inicializador do server com as seguintes seções:

[ONSTART]


Jobs=WORKFLOW

[WORKFLOW]


Main=WFONSTRT

Environment=Enviromment


Para agendarmos as tarefas (Jobs) precisamos entrar no configurador.



Selecione: **"Processos Especiais"**.



Clique em **Incluir**



## Scheduler – Processos Especiais

Agrupamento	Campos	Descrição
<b>Identificação</b>	Código	Código único identificador do <i>job</i> .
	Nome	Nome simplificado.
	Descrição	Descrição do <i>job</i> .
<b>Frequência</b>	Diária	Executa diariamente entre os horários inicial e final a cada intervalo de tempo.
	Semanal	Executa nos dias selecionados da semana (domingo, segunda, terça, quarta, quinta, sexta e sábado), entre os horários inicial e final a cada intervalo de tempo.
	Mensal	Executa nos meses selecionados do ano (janeiro a dezembro), entre os horários inicial e final a cada intervalo de tempo.
<b>Período Inicial</b>	Data Início	Data inicial que passará a vigorar a execução.
	Hora Início	Horário inicial de execução.
<b>Período Final</b>	Data final	Dia limite de execução.
	Hora final	Horário limite de execução.
<b>Período</b>	Intervalo	Intervalo de tempo a cada execução entre o horário inicial e final.
<b>Job</b>	Ação	Função de usuário ou função interna do Protheus a ser executado. Os parâmetros passados para esta função deverão ser sempre tratados como um <i>array</i> .
<b>Environment</b>	Ambiente	Ambiente em que o <i>job</i> será executado.

É aconselhável configurar o recebimento automático de tempos em tempos caso haja algum problema de retardo de email ou gargalo de fluxo de email.

**Inclusão**

Pasta: Processos Especiais

Processos Especiais

Código	Nome	Descrição
WFSND	envia	ENVIA

Frequência

☒ Diária  
☐ Semanal  
☐ Mensal

Data Início: 23/05/07 ?  
Data Final: 23/05/07 ?

Hora Início: 00:00  
Hora Final: 23:59

Intervalo: 00:05

Ação: WFSendMail( "99", "01" )

Ambiente: ENVIRONMENT



## Criação do HTML

Você deve criar um formulário em html para enviar no email pelo workflow, veja exemplo:

```
<!DOCTYPE HTML PUBLIC "-//W3C//DTD HTML 4.01 Transitional//EN">
<html>
<head>
<title>Untitled Document</title>
<meta http-equiv="Content-Type" content="text/html; charset=iso-8859-1">
</head>

<body>
<p><font size="2" face="Verdana, Arial">Prezado Aprovador!</font></p>
<p><font size="2" face="Verdana, Arial">Ap&ocirc;e; s o lan&ccedil;amento abaixo,
o saldo de <strong>!NOME!</strong> ficou negativo em <strong>R$ !SALDO!</strong>
.</font></p>
<font size="2" face="Verdana, Arial">Nr. Transa&ccedil;&atilde;o:
<strong>!NUMERO!</strong></font>
<br><font size="2" face="Verdana, Arial">Item: <strong>!ITEM!</strong></font>
<br><font size="2" face="Verdana, Arial">Data: <strong>!DATA!</strong></font>
<br><font size="2" face="Verdana, Arial">Hist&ocirc;e;rico:
<strong>!HIST!</strong></font>
<br><font size="2" face="Verdana, Arial">Valor: <strong>!VALOR!</strong></font>
<form name="form1" method="post" action="mailto:%WFMailTo%">
  <p><font size="2" face="Verdana, Arial"><input type="radio" name="%APROVA%"
value="Sim">Aprova este lan&ccedil;amento</font></p>
  <p><font size="2" face="Verdana, Arial"><input type="radio" name="%APROVA%"
value="N&atilde;o">N&atilde;o aprova</font></p>
  <p><font size="2" face="Verdana, Arial"><input type="submit" name="Submit"
value="Enviar"></font> </p>
</form>
</body>
</html>
```

A rotina de workflow substitui todas as palavras encontradas nos formulários html que estiverem entre "%" ou "!". Essa palavra-chave terá seu conteúdo modificado por outra informação durante a execução do processo.

## Criação de Processos em ADVPL

Um processo de Workflow poderá ser iniciado a partir de uma destas opções:

- pontos de entrada;
- item de menu de qualquer ambiente;
- *job* agendado através do Scheduler;
- manualmente através do remote.

Na maior parte, o Workflow baseia-se, no uso de classes referenciadas aos objetos, que se tornam a ponte de acesso para uso de seus métodos e propriedades. A principal classe do Workflow é a **"TWFPProcess"**, o que veremos a seguir (é necessário conhecimento de programação orientada a objetos).

## Classe TWFPPROCESS

A classe TWFPProcess é responsável pela criação e gerenciamento do processo. Como toda classe, a TWFPProcess é dividida em métodos e propriedades. Veremos alguns dos principais métodos e propriedades que iremos usar para criação de um processo.

### MÉTODOS

- **NEW** - O método New() é responsável pela criação e inicialização da classe WFProcess.

Sintaxe

:New(<cCodProc>,<cDescr>,<cProcID>)

Argumento	Obrig	Tipo	Descrição
cCodProc	Sim	C	Recebe o código do processo usado em "Cadastro de Processos".
cDescr	Sim	C	Recebe a descrição do processo que está sendo criado no momento. Se não for informado, será usada a descrição contida no cadastro de processo, localizada através do parâmetro anterior cCodProc
cProcID	Não	C	Recebe o ID do processo criado anteriormente. Normalmente, é utilizado para reconstruir um processo anterior, dando seqüência a ele

### Exemplo:

oP := TWFPProcess():New("PEDCOM","Aprovação do Pedido de Compras")

- **NEWTASK** - Responsável em criar a sequência de tarefas a ser executada e identificar o html que será utilizado.

Sintaxe

:NewTask(<cDescr>, <cArqHtml>, <lCopiar>)

Argumento	Obrig	Tipo	Descrição
cDescr	Sim	C	Recebe a descrição da tarefa.
cArqHtml	Sim	C	Recebe o caminho e o nome do arquivo html que fará uso no processo.
lCopiar	Não	L	Copia os campos utilizados na tarefa anterior para a tarefa que está sendo criada.

**Exemplo:**

oP:NewTask("Criando Aprovação", "\Workflow\WFW120p.htm" )

- **ATTACHFILE** - Responsável pelos arquivos anexos à mensagem, que deverão estar dentro do rootpath do Protheus.

Sintaxe

:AttachFile(<cArquivo>)

Argumento	Obrig	Tipo	Descrição
cArquivo	Sim	C	Caminho e nome do arquivo a ser anexo à mensagem

**Exemplo:**

oP:AttachFile("\Workflow\teste.txt")

- **START** - Responsável por construir o processo, gravar nas tabelas do Workflow e enviar mensagem aos destinatários. O retorno é uma chave composta por **ProcessID+TaskID** composto por 17 algarismos hexadecimais.

Sintaxe

:Start(<cHtmlCopiarPara>) -> cProcessKey

Argumento	Obrig	Tipo	Descrição
cHtmlCopiarPara	Sim	C	Caminho em que o Workflow deverá realizar uma cópia do html final.

**Exemplo:**

```
cID := oP:Start("\Workflow\copia")
if file( "\Workflow\copia\" + cID)
    conout("Arquivo copiado com sucesso.")
endif
```

**FINISH** - Este método é responsável por finalizar o processo. Após a finalização, ele não estará mais disponível para execuções do tipo retorno e timeout.

Sintaxe

:Finish()

**Exemplo:**

oP:Finish()

- **TRACK** - Responsável por incluir as descrições dos passos seguidos pelo fluxo do processo e apresentá-los na consulta da rastreabilidade.

Sintaxe

:Track(<cCodStatus>,<cDescr>,<cUsuario>)

Argumento	Obrig	Tipo	Descrição
cCodStatus	Sim	C	código do status do processo.
cDescr	Sim	C	descrição do passo ocorrido.
cUsuario	Sim	C	nome do usuário a que se destinou a tarefa.

**Exemplo:**

oP:Track("100200", "Enviando o pedido para aprovação", "AprovadorA")

## PROPRIEDADES

**:cTo, :cCC e :cBCC**

Estas propriedades definem o endereço dos destinatários. Para informar mais que um destinatário, basta incluir um ponto-e-vírgula ";" entre eles. Se for informada uma palavra qualquer que não seja um endereço de e-mail válido, o Workflow considera que é um diretório e deverá gerar o HTML. Pode-se mesclar os tipos.

**Exemplo:**

oP:cTo := "aluno1@advpl.com.br;aluno2@advpl.com.br"

oP:cCC := "aluno3@advpl.com.br;Aluno"

**oP:cBCC := "aluno4@advpl.com.br"**

**:cSubject**

Esta propriedade define o assunto da mensagem.

**Exemplo:**

oP:cSubject := "Aprovado do pedido de compras no. 1028"

**:cBody**

Esta propriedade armazenará um texto que permanecerá no corpo da mensagem. Caso seja utilizado, o html irá como anexo da mensagem.

**Exemplo:**

oP:cBody := "Testando..."

**:bReturn**

Esta propriedade contém o nome da função que será executada no momento em que o Workflow receber a mensagem de resposta de um dos destinatários via e-mail ou serviço http.

**Exemplo:**

oP:bReturn := "U\_Returno"

**:bTimeOut**

Esta propriedade recebe um *array* de *timeouts* contendo nomes das funções e tempo de espera. Caso o tempo seja alcançado, serão executadas as funções mencionadas no 1º item do *array*. Poderão ser especificados mais do que um *array* de *timeouts*.

Sintaxe:

{ { <cFuncao>, <nDias>, <nHoras>, <nMinutos> }, { ... } }

**Exemplo:**

oP:bTimeOut := { { "TimeOut1", 0, 5, 30 } }

ou

oP:bTimeOut := { { "TimeOut1", 0, 5, 30 }, { "TimeOut2", 1, 10, 0 } }

**:fProcessID**

Esta propriedade fornece o número ID do processo.

**Exemplo:**

cProcID := oP:fProcessID

**:fTaskID**

Esta propriedade fornece o número ID da tarefa criada para um determinado processo, através do método **:NewTask()**.

**Exemplo:**

oP:NewTask( "100100", "\\Workflow\WFW120p.htm" )

cTaskID := oP:fTaskID

**:oHTML**

Esta propriedade é responsável pelo tratamento das palavras-chaves no html mencionado no método **:NewTask()**. Esse objeto é uma referência da classe TWfHtml.

**CLASSE TWfHTML**

A classe TWfhtml é responsável pela leitura do HTML de retorno e de envio para tratamento das variáveis Veremos alguns dos principais métodos e propriedades que iremos usar.

**MÉTODOS**

- **VALBYNAME** - Este método tem como objetivo atribuir ou obter um valor a uma "macro" existente no html. Deverá ser usado somente no momento em que estiver enviando workflow ou na função de timeouts, em que o uso é necessário por motivo de não haver recebido resposta.

### :oHTML:ValByName(<cMacro>,<uConteudo>)

Argumento	Obrig	Tipo	Descrição
cMacro	Sim	C	nome da macro (palavra-chave) encontrada no html e identificada entre os símbolos "%" e "!".
uConteudo	Sim	C	valor a ser atribuído à macro.

#### Exemplo:

##### Assinalando um valor:

oP:oHtml:ValByName( "Nome", "Aluno1" )

##### Usando na função de *timeout*:

cNome := oP:oHtml:ValByName( "Nome" )

##### Atribuindo um valor a uma tabela:

AAdd( oP:oHtml:ValByName( "produto.codigo" ), SB1->SB1\_COD )

- **RETBYNAM** - O método retbyname() é responsável pelo tratamento das variáveis de retorno do workflow. O método somente deverá ser usado na **função de retorno**.

Sintaxe:

:oHTML:RetByName( cMacro )

Argumento	Obrig	Tipo	Descrição
cMacro	Sim	C	nome da macro (palavra-chave) encontrada no html e identificada entre os símbolos "%" e "!" no formulário de retorno.

#### Exemplo:

cNome := oP:oHtml:RetByName( "Nome" )

##### Em uma tabela:


aCodigo := oP:oHtml:RetByName( "produto.Codigo" )

## Exercício

### 3. Criar um workflow de clientes:


- Criar um campo: A1\_AU\_REVI C 1 COMBO 1=SIM;2=NÃO sem inicializador padrão somente visual
- Alterar o campo A1\_MSBLQL para visual e colocar inicializador = 1
- Usar o ponto de entrada M030INC para enviar um workflow para o cliente(A1\_EMAIL) com suas informações cadastrais.
- Tratar o retorno da seguinte forma:
  - a. Resposta = sim,
    - ✓ alterar A1\_AU\_REVI = 1 (sim) e A1\_MSBLQ = 2 (não)
  - b. Resposta = não,
    - ✓ alterar A1\_AU\_REVI = 1 (sim) e A1\_MSBLQ = 1 (sim)

Montar Email da Seguinte forma:



Untitled Document - Microsoft Internet Explorer

Arquivo Editar Exibir Favoritos Ferramentas Ajuda

Endereço  Ir Links >> Google  Pesquisar >> Fazer

Prezado Cliente!

Seu cadastro de cliente precisa ser liberado para ser usado no sistema!

Seu código: **000004**  
Nome: **TESTE DE INCLUSAO**  
CPF:  
Endereco: **TESTE**  
Telefone: **99 9999 9999**  
CEP: **99999-999**  
Bairro: **TESTE**  
Município: **TESTE**  
Estado: **ZZ**

☐ Aprova essa modificação!  
☐ Não aprovo essa modificação!

Concluído Meu computador





## Capítulo 05 – WEB Service

Quando um Web Service é criado e disponibilizado, junto dele também é disponibilizada a definição do serviço, seus argumentos, estruturas e retornos (WSDL). Para nos utilizarmos de um Web Service, precisamos montar um programa - Client que seja capaz de montar um "envelope" SOAP com os dados necessários ao processamento do Serviço, realizar a chamada, e tratar o respectivo retorno e exceções.

Embora existam Web Services que podem ser acessados via Http "direto", apenas passando parâmetros via URL, o Client de Web Services do Advanced Protheus tem seu foco e recursos direcionados apenas à serviços que possuam interface de comunicação que realize POST de pacotes de dados XML em formato SOAP.

Para a codificação de um webservice, foram criadas em Advpl instruções especiais de declaração de classe, específicas para WebServices, que suportam nomes 'longos' no nome da classe, métodos e propriedades. A utilização destes comandos exige a declaração do `#include 'apwebsrv.ch'` no topo do código-fonte; e exige também a atenção em alguns pontos e particularidades, a iniciar pela nomenclatura do serviço, estruturas, métodos e propriedades.

### TERMOS TÉCNICOS

**SOAP** - Sigla de Simple Object Access Protocol, ou protocolo simples de acesso a objetos. O SOAP é um padrão aberto, baseado em XML, criado pela Microsoft, Ariba e IBM para padronizar a transferência de dados entre aplicações. Pode ser usado em combinação com vários outros protocolos comuns da Internet, como HTTP e SMTP.

**WSDL** - Web Service Description Language. Submetida à W3C - o órgão padronizador da Internet - . A linguagem WSDL define regras baseadas em XML para descrever serviços web.

**WEB SERVICES** - programa completo ou componente de software residente num servidor web.

**XML** - Sigla de Extensible Markup Language, o XML é uma linguagem baseada em tags semelhante ao HTML. Sua principal característica é a extensibilidade. Quem emite um documento XML pode criar tags personalizadas, que são explicadas num documento anexo, que tem extensão XSD.

**XSD** - Sigla de XML Schema Definition. Arquivo associado a um documento XML que descreve e valida aos dados no documento. Assim como as linguagens de programação, os XSDs aceitam dados de diferentes tipos, como números, data e moeda.

### DEFINIÇÃO DO CLIENT

Para a criação do código-fonte de um Client de um Web Service, foi criada uma ferramenta de geração automática do fonte a partir da definição do Serviço (WSDL), integrada ao IDE, e foram criadas classes especiais reservadas em Advpl para a montagem de tal fonte.

Ao gerarmos um fonte client para um WEB Service, este conterá as definições dos métodos / ações do serviço , a(s) estrutura(s) utilizada(s) no mesmo , e a(s) classe(s) intermediária(s) de uso interno para montagem e desmontagem da(s) estrutura(s) ; visando o encapsulamento de todos os tratamentos de envio e recebimento de dados através de pacotes SOAP.


O Fonte gerado através do assistente de criação de fonte deve preferencialmente ser gerado e compilado em um arquivo exclusivo , destinado unica e exclusivamente a este código. E , por tratar-se de uma classe Advpl gerada a partir da definição de um serviço , não devemos inserir e/ou alterar nenhuma das definições geradas pelo assistente , pois as mesmas serão perdidas caso o fonte seja gerado novamente .

## Configuração

Tanto WebService quanto ADVPL ASP devem obrigatoriamente ser configurados pelo Wizard, pois são descompactados pacotes necessários para o perfeito funcionamento da rotina.

Passos:

- Configurar o HTTP



### Configurações da Seção

☒ Protocolo Habilitado

Nome da Instância

Path de Arquivos

Porta de Conexão

Ambiente

Processo de Resposta


Instâncias do Protocolo (mínimo)

Instâncias do Protocolo (máximo)

Path para Upload de Arquivos

Time-Out de Sessions WEBEX (em segundos)

## - Configurar o serviço Web service



Pasta: Módulos Web\ws


Ambientes
   
 Servidor de Licenças
   
 Módulos Web
   
 ws
   
 Servidor Internet (HTTP/FTP)
   
 Hosts / URLs HTTP
   
 Processos Comuns
   
 Processos WEB / WEBEX
   
 Conexão (Server/Remote)
   
 TOTVSDBAccess
   
 Servidor CTREE
   
 Balanceamento de Carga
   
 Serviço do Windows NT/2000
   
 Geral

Tipo do Módulo  
 WS - Protheus 10 Web Services
   
 Nome da Instância  
 ws
   
 Diretório inicial do ambiente Web  
 C:\Aula\ADVPL\Protheus\_Data\web\ws
   
 Ambiente utilizado  
 CRIS
   
 URLs de Acesso (Hosts)
 

Host	Empresa/Filial	Min. Usuários	Max. Usuários
192.168.0.114:8080/ws	9901	10	30

  
 Working Threads ( Jobs )
 

Job	Empresa/Filial	Min. W.Thread	Max. W. Thread
192.168.0.114:8080/ws	9901	10	30



**Assistente de Configuração de Módulo Web - Novo módulo**

**Configurações Básicas do Módulo**

Módulo Web  
WS - Protheus 10 Web Services

Nome da Instância  
ws

Diretório Raiz das imagens ( Web Path )  
\\web\ws

Selecione o Environment  
ENVIRONMENT


☒ Habilitar processos na Inicialização do Servidor

< Voltar Avançar >> Cancelar

Clicar em cada empresa e relacionar ao Host

Se a porta do http for diferente de 80, voce deve colocar na configuração do WS da seguinte forma:

Ex. porta 8080 - localhost:8080/ws



**Assistente de Configuração de Módulo Web - Editando Instância : [ws]**

**Configuração de Hosts X Empresas/filiais**  
Definindo o host e as empresas

Host ( Pode incluir o diretório virtual )  
localhost:8080/ws  
[HTTP] DEFAULT HOST  
192.168.0.114:8080/ws

Selecione a Empresa/Filial  
9901 - TESTE / MATRIZ

Relacionar Excluir

Relacionamentos

Host	Empresa/Filial
192.168.0.114:8080/ws	9901 - TESTE / MATRIZ

<< Voltar Avançar >> Cancelar

**Assistente de Configuração de Módulo Web - Editando Instância : [ws]**

**Configuração de usuários**  
Definição de expectativa de acessos

Host Virtual	Empresa/Filial	Mínimo Usuários	Máximo Usuários
192.168.0.114:8080/ws	9901 - TESTE / MATRIZ	10	30

<< Voltar Finalizar Cancelar

- Testar o serviço

Reinicializar o server

```

AppServer
----- OS Memory Info -----
Physical memory .      2910.88 Mb.   Used      1102.77 Mb.   Free       1808.11 Mb.
Paging file .....    4791.59 Mb.   Used      1163.04 Mb.   Free       3628.55 Mb.
Virtual memory ..     2047.88 Mb.   Used       33.57 Mb.   Free       2014.30 Mb.

Http server is ready.
  Root path is c:\protheus10help\
  Listening port 8080


[INFO ][SERVER] TOTVS Application Server started on port 7892
[29/06/2010 16:05:31] Server started

Starting Job [WORKFLOW_ADUPL]

Starting Job [JOB_WS_9901]
*** Loading totvsdbapi.dll

Loading Ctree Local [ctreestd.dll]
*** Loading ace32.dll - Version: 7.00 (7.0.0.5)
*** Loading totvsdbapi.dll
WARNING: Servidor de licenças não está respondendo.
  
```

No Internet Explorer: **http://localhost:8080/ws** ou o ip **<http://192.168.0.106:8080/ws>**



## - Criar um Web service

- **WsService** -> Cria o WebService

Ex: WsService ForneceSaldo Description "Fornece Saldo da conta"

- **WsData** -> Definição das variáveis, como String, float, Array of String

Ex: WsData Nome As String

- **EndWsService** -> finaliza o WebService
- **WsMethod** -> Cria os métodos do WebService

WsMethod BuscaSaldo Description "Busca o saldo da conta."

**WsReceive** -> Define variavel com conteúdo a ser pesquisado.

**WsSend** -> Define variavel com conteúdo que será devolvido.

Ex: WsMethod BuscaSaldo WsReceive Nome WsSend Saldo WsService ForneceSaldo

A programação é feita normalmente em ADVPL, exceto na declaração e atribuição de variáveis, que deve ser precedida pelo símbolo ( :: ) quando a variável definida como WSDData receber algum tipo de atribuição:

::Saldo := 0

#### Exemplo de Código para criação de Web Service

```
#INCLUDE "PROTHEUS.CH"
#INCLUDE "APWEBSRV.CH"
WsService ForneceSaldo Description "Fornece o saldo de uma conta."
  WsData Nome          As String
  WsData Saldo          As Float
  WsData Transacoes As Array Of string

  WsMethod BuscaSaldo      Description "Busca o saldo da conta."
  WsMethod BuscaTransacoes Description "Busca as contas e as suas transacoes."
EndWsService

WsMethod BuscaSaldo WsReceive Nome WsSend Saldo WsService ForneceSaldo
  ::Saldo := 0
  SZ1->(dbOrderNickName("NOME"))
  If SZ1->(dbSeek(xFilial("SZ1")+ Nome))
    ::Saldo := SZ1->Z1_Saldo
  EndIf

Return .T.

WsMethod BuscaTransacoes WsReceive NULLPARAM WsSend Transacoes WsService ForneceSaldo
  ::Transacoes := {}
  SZ1->(dbOrderNickName("NOME"))
  SZ1->(dbSeek(xFilial("SZ1")))

  While SZ1->Z1_Filial == xFilial("SZ1") .And. !SZ1->(Eof())// Le o cadastro de contas.
    AAdd(::Transacoes, "SZ1" + SZ1->Z1_Nome)


    SZ2->(dbOrderNickName("NOME_NR_IT"))
    SZ2->(dbSeek(xFilial("SZ2") + SZ1->Z1_Nome))

    While SZ2->Z2_Filial + SZ2->Z2_Nome == xFilial("SZ2") + SZ1->Z1_Nome .And. !SZ2->(Eof())
      AAdd(::Transacoes, "SZ2"+DtoC(SZ2->Z2_Data)+SZ2->Z2_Tipo+Transform(SZ2->Z2_Valor, "@E 999,999,999.99"))
      SZ2->(dbSkip())
    End
    SZ1->(dbSkip())
  End

Return .T.
```

- Criar um client para Web Service Criado


Utilizar o Editor de Fontes Development Studio para gerar o client a partir de uma WSDL



- o caminho completo deve ser fornecido pelo Fornecedor do WEB Service será parecido com o seguinte:


<http://localhost:8080/ws/9901/FORNECESALDO.apw?WSDL>

devera ser digitado na seguinte tela:





Clicar em OK e o IDE irá gerar o código automaticamente, bastando apenas salvar, inserir no projeto e compilar, assim a Classe referente a esse Serviço já estará disponível para ser utilizada em qualquer programa nesse ambiente.



- Testar o web service criado.

Necessário criar um programa em ADVPL que faz a criação do Objeto a partir da Classe gerada pelo IDE.

#### Exemplo de um Client acessando um WS

```
Static Function Saldo(cNome, oSaldo, nSaldo)

Local oWS      := WSForneceSaldo():New()
Local nSaldo := 0

If oWS:BuscaSaldo(cNome)
    nSaldo := oWS:nBuscaSaldoRESULT
Else
    MsgStop("WSForneceSaldo: Web Service nao acessivel!")
EndIf

oSaldo:SetText(nSaldo)
oSaldo:Refresh()

Return .T.
```

## Exercício

4. Criar um Web Service de Saldo de Produtos:

-Recebe:Código do Produto (B1\_COD)

-Retorno de acordo com 3 métodos

1. Saldo do Produto (B2\_QATU)

2. Saldo do Produto m Valor (B2\_VATU1)

3. Quantidade Disponível (B2\_QATU-B2\_RESERVA-B2\_QEMP)

Criar um Cliente para esse Web Service para testá-lo usando os 3 métodos.

## Capítulo 06 – ADVPL ASP

Uma página ASP (Active Server Pages) é uma página HTML contendo código interpretável em uma linguagem compreensível ao servidor HTTP em uso. Por exemplo, o IIS da Microsoft utiliza o VBScript para criar suas páginas ASP, do mesmo modo que o Protheus utiliza o ADVPL. Uma página ASP é uma combinação de script HTML e código interpretável. No ADVPL ASP esse código é padrão xBase, portanto a preocupação maior daqueles que já conhecem e trabalham com o Protheus e desejam desenvolver páginas ativas para aplicações Web utilizando essa facilidade é conhecer HTML.

### Características do ADVPL ASP – Arquivos .APH

Os arquivos ADVPL ASP têm a extensão padrão .APH. São arquivos texto e devem ser adicionados a um projeto no Protheus IDE e compilados da mesma maneira que os programas tradicionais. A diferença é que o Protheus Server identificará que se trata de um ADVPL ASP e executará uma espécie de tradutor (parser) antes que a compilação seja executada. Este parser irá transformar todo o arquivo em uma função única, que receberá os mesmos parâmetros das funções APL simples, como explicado anteriormente no Item 'Desenvolvendo Funções. APL', e retornará uma string.

O desenvolvedor não precisa se preocupar em retornar HTML algum, pois o APH também é um arquivo HTML. A função que foi gerada na compilação irá se encarregar de retornar o HTML contigo no arquivo, depois que o código foi processado. Um Arquivo APH gera no repositório de Objetos do Protheus uma função com o mesmo nome do arquivo, porém prefixada com H\_ + nome\_do\_arquivo\_apl

Por se tornar uma função no momento da compilação, não é possível utilizar a cláusula FUNCTION para criar outras funções dentro de um arquivo APH. Caso exista essa necessidade, tais funções devem ser criadas em um arquivo PRW tradicional e chamadas de dentro do APH.

A extensão APH para o nome dos arquivos é obrigatória. Qualquer outra extensão usada no nome do arquivo não será reconhecida e o parser do IDE não será executado durante a compilação. Foi criada também a extensão de arquivos. AHU ( Aph de Usuário ), que possui o mesmo tratamento de Parser do arquivo APH, gerando uma função prefixada com L\_. A diferença é que a função gerada pelo AHU pode ser gerada sem a necessidade de autorização de compilação com permissão para substituir fontes Microsiga, por tratar-se de um Aph de Usuário, equivalente à uma User Function.

Assim como outros ASP's, a diferenciação entre script HTML e código é efetuada através dos caracteres <% para indicação de abertura de código e %> para indicação do encerramento de código. Por exemplo, o HTML abaixo contém um pedaço de código ADVPL separado pelos delimitadores:

```
<html><head><title>ADVPL ASP Demo</title></head>
<body>
```

<p>Bem vindo ao mundo do ADVPL ASP!</p>

<%

// Soma os 100 primeiros números

Local i, nSoma := 0

For i := 1 To 100

    nSoma += i

Next i

%>

</body>

</html>

Quando este arquivo for requisitado ao Protheus Server (através de uma chamada em URL, por exemplo) o código entre os delimitadores será executado, porém o script colocado ao redor do código será mantido exatamente como se encontra. Por exemplo:

**http://localhost/H\_WEBDEMO.APL**

A grande vantagem de se utilizar dos arquivos ADVPL ASP em relação a criar funções APL simples, decorre do fato de que nas funções APL simples o desenvolvedor deve se preocupar em retornar todo o HTML necessário para a correta exibição no Web Browser.

E também, como o ADVPL ASP mistura o script HTML com o código interpretável, pode-se criar um arquivo APH utilizando o editor desejado (como o Microsoft FrontPage, por exemplo) e inserir nele os códigos Advpl necessários entre as Tags. Outro detalhe importante é que pode-se utilizar as estruturas de fluxo da linguagem ADVPL para repetir comandos do próprio script HTML (por exemplo, colocar um comando de script HTML dentro de um comando While em ADVPL):

<% While !EOF() %>

<B> Esta linha será repetida no HTML até ocorrer o fim de arquivo </B>

<%

dbSkip()

    EndDo

%>

Note que também podem existir diferentes blocos de código interpretável separados pelos delimitadores, dentro de um mesmo arquivo.

Tão importante quanto mesclar código interpretável com script de formatação HTML, é utilizar os comandos ADVPL para alterar o script de formatação. Ou seja, colocar o conteúdo de variáveis, campos, etc, diretamente no HTML que será enviado à aplicação client (ao Browser por exemplo). Isso pode ser realizado através dos delimitadores de avaliação. Os delimitadores de avaliação são **<%=** para abertura e **%>** para encerramento. Diferentemente dos delimitadores de código interpretável, estes devem sempre estar na mesma linha. Com eles pode-se criar uma linha de script HTML, cujo conteúdo contém uma expressão que será avaliada em tempo de execução, e seu resultado inserido como parte do HTML retornado ao Browser:

<b>Esta linha é HTML, mas o horário exibido aqui: <%= Time() %> foi obtido em tempo de execução no Servidor.</b>

No exemplo acima, a linha HTML será retornada para o Browser com o resultado da função time (ou seja, a hora atual no servidor) inserido no texto.

Utilizando todos esses conceitos, pode-se criar toda uma aplicação Web baseada no Protheus. Ou seja, o processamento e acesso aos dados ficam por conta do Protheus Server, e a interface fica por conta do Browser (utilizando o HTML) .

### Importante

Ao codificar um arquivo .APH e/ou .AHU , devemos estar atentos às regras de utilização dos delimitadores de execução e avaliação Advpl :

1. A Abertura e fechamento dos delimitadores de execução <% ... %> devem estar isoladas em suas respectivas linhas, não devendo ter qualquer conteúdo adicional, nem duas aberturas e fechamentos na mesma linha. Partindo dessa premissa, veja abaixo alguns exemplos de como inserir o código Advpl abaixo dentro de um APH:

#### CERTO

```
<% IF !IOk %>
<% nErro++ %>
<% Endif %>
```

```
IF !IOk
nErro++
Endif
```

```
<% IF !IOk ; nErro++ ;
Endif %>
```

```
<%
IF !IOk
nErro++
Endif
%>
```

## ERRADO

```
<% IF !IOk %><% nErro++ %> -- 2 aberturas e fechamentos na mesma linha  
<% Endif %>
```

2. Quanto aos delimitadores de avaliação `<%= ... %>`, podemos ter várias aberturas e fechamentos na mesma linha, porém não podemos ter uma abertura e seu respectivo fechamento em linhas diferentes.

3. Uma linha qualquer em um arquivo .APL não deve conter mais do que 150 Caracteres, pois o Parser insere caracteres de controle em cada linha do mesmo durante a pré-compilação. e a linha final resultante não pode ultrapassar 254 caracteres, pois neste caso isto impossibilita a compilação do APL.

## Desenvolvimento de Funções .APL

A princípio, todas as funções contidas no repositório podem ser executadas diretamente através de uma requisição HTTP, via link com extensão .apl, ao Protheus Server. Porém, alguns detalhes devem ser considerados:

- Uma função executada no momento do recebimento de uma requisição HTTP é executada como um JOB, ou seja, não contém interface. Por isso, tais funções não podem conter comandos de interface, como criação de janelas ou exibição de helps e mensagens de alerta;
- A única interface possível é a utilizada no client HTTP. Por isso, tais funções devem SEMPRE retornar uma string de caracteres. Após o processamento da função, essa string de retorno será enviada diretamente ao client HTTP e este será o responsável por sua interpretação. Por exemplo, utilizando um Web Browser como client pode-se retornar a string de comandos HTML diretamente. O HTML então será propriamente exibido no Web Browser;
- Qualquer retorno diferente de uma string de caracteres gerará um erro que será enviado à aplicação client HTTP (o erro gerado é ?Invalid Proc Return?);
- O servidor Protheus passa alguns parâmetros para as funções chamadas. Isso significa que ao criar funções para serem utilizadas em resposta às requisições HTTP, deve-se criar o cabeçalho da função com estes parâmetros. Não é obrigatório utilizar os mesmos nomes de parâmetros sugeridos abaixo quando criar diretamente estas funções. Porém, como são esses os nomes utilizados no ADVPL ASP explicado mas a frente, é aconselhável utilizá-los por motivo de padronização:

\_\_\_aCookies: Este parâmetro recebe um array bidimensional com os Cookies criados na aplicação client HTTP (por exemplo, no Internet Explorer 5). Pode-se utilizá-lo para checar validações mantidas nas máquinas client por exemplo. Para maiores detalhes, consulte a documentação do HTML ou do Web Browser utilizado.

◦ `__aPostParms`: Este parâmetro recebe um array bidimensional com os campos contidos em um formulário HTML recebido através de um comando POST. Cada item deste array contém um array com o nome do campo e o valor informado. Por exemplo, para um formulário com dois campos para digitação (um chamado nome e o outro chamado endereço), que enviam os dados para a função `cadastro.apl` através de um POST, a função receberá o array `__aPostParms` da seguinte forma:

```
{ {?nome?, ?NOME DIGITADO NA PAGINA HTML?},  
  {?endereco?, ?ENDERECO DIGITADO NA PAGINA HTML?} }
```

A função pode tratar os dados recebidos neste array para realizar um processamento específico com tais informações. Para campos onde não é possível a entrada de dados e sim a escolha de uma informação pré-definida (como por exemplo um checkbox), o item somente existirá no array caso o campo tenha sido selecionado no formulário HTML (por exemplo, se o checkbox for marcado).

◦ `__nProcID`: Contém o Handle da Thread de execução daquela função. A utilização deste parâmetro será explicada juntamente com o tópico ADVPL ASP posteriormente;

◦ `__aProcParms`: Este parâmetro recebe um array bidimensional com os parâmetros enviados na linha de URL do Web Browser. Por exemplo, na execução de uma função via linha de URL do Web Browser como:

```
http://servidor/vende.apl?cod=000001&nome=PRODUTO DE  
TESTE&quant=20  
a função chamada vende receberá o array __aProcParms da seguinte  
forma:
```

```
{ {?cod?, ?000001?},  
  {?nome?, ?PRODUTO DE TESTE?},  
  {?quant?, ?20?} }
```

A função pode tratar estes dados recebidos para realizar processamentos específicos. É muito útil também para criar links de execução diretamente através de um Web Browser.

◦ `__cHTTPPage`: Esse parâmetro foi criado originariamente para receber o nome da página, arquivo ou função originalmente requisitada para o Protheus Server, porém não foi utilizado e permaneceu por compatibilidade. Caso consultado, ele retorna uma string em branco.

◦ `__aHTTPHead`: Esse parâmetro recebe um array com os Headers do cabeçalho da requisição HTTP enviados pelo Web Browser.

### Configuração Mínima

Em tópico à parte é explicada toda a configuração referente à seção http do Protheus Server. A configuração abaixo é a mínima necessária para executar o exemplo acima

[http]

Port=80

Path=(caminho absoluto de disco para arquivos publicados no servidor )

Environment=(nome do environment do Serber que será utilizado para o processamento)

Para testar sua configuração, reinicie o server Protheus e chame via WebBrowser a url :

<http://localhost/time.apl>

Deverá ser mostrada no Browse o horário atual, no formato hh:mm:ss, no Servidor, pois este é o resultado da função Advpl TIME() .

O Protheus atendendo à requisições .apl

Quando solicitado através de um Web Browser um processamento de uma função via link .apl, a função solicitada deve ser responsável por abrir o ambiente necessário ao processamento, conectar com o Banco de Dados caso necessário, realizar o processamento e retornar a String Html ao Web Browser.

Este ambiente criado é fechado imediatamente após o término do processamento, o que exige muito do Servidor da aplicação em se tratando de uma aplicação web com vários usuários efetuando acessos simultâneos. Para atender com mais eficiência às requisições de processamento de um projeto web, foi implementada no Protheus a tecnologia de 'Working Threads', explicada em mais detalhes no tópico 'Desenvolvimento de Funções .apw'.

#### Exemplo de um PRW acessando um Portal

```
#Include "Protheus.ch"
#Include "APWEBEX.ch"
#Include "TbiConn.ch"
User Function Inicio()

Local cHtml := ""
    WEB EXTENDED INIT cHtml Start "u_VldInicio"

    HttpSession->dData := Date()
    HttpSession->cHora := Time()
    cHtml += ExecInPage("INICIO")

    WEB EXTENDED END

Return cHtml

User Function Menu()

Local cHtml := ""
Local cOpc

    WEB EXTENDED INIT cHtml
```

Continua...



Continuação...

```
cOpc := HttpGet->Opc
If cOpc == "1"
    u_Opc1()
EndIf

WEB EXTENDED END

Return cHtml

User Function Opc1()

Local cHtml := ""

    WEB EXTENDED INIT cHtml

    Prepare Environment Empresa "99" Filial "01" Modulo "FAT" Tables "SA1"
    cHtml += ExecInPage("OPC1") //PAGINA DA LISTA DE CLIENTES
    Reset Environment

    WEB EXTENDED END

Return cHtml
```

**Exemplo de um APH acessando um Portal – INICIO.APH**

```
<html><head><title>Menu</title></head>

<body>

<table width="100%" border="0" cellspacing="0" cellpadding="0">
<tr>
<td bgcolor="#0066CC">
<font color="#CCCCCC" size="5" face="Verdana">
<em>&nbsp;Portal de vendas Junior Japa </em> </font>
</td></tr>
<tr><td bgcolor="#003366">
<div align="right"><strong>
<font color="#E9E9E9" size="2" face="Verdana">
<%=HttpSession->dData%> - <%=Left(HttpSession->cHora,5)%>&nbsp;</font>
</strong></div></td></tr>
</table>
<p align="left"><a href="u_Menu.apw?Opc=1">
<font size="2" face="Verdana">Lista de clientes </font></a></p>

</body>

</html>
```

#### Exemplo de um APH acessando um Portal – OPC1.APH

```
<html><head> <title>Lista de Clientes</title></head>

<body>

<table width="100%" border="2" cellpadding="3" cellspacing="5" bordercolor="#FFFFFF"
bgcolor="#CCCCCC">

<td align="center" bgcolor="#666666"><font color="#CCCCCC" size="2"
face="Verdana"><strong>Codigo</strong></font></td>
<td align="center" bgcolor="#666666"><font color="#CCCCCC" size="2"
face="Verdana"><strong>Nome</strong></font></td>

<%DbSelectArea("SA1")%>
<%SA1->(DbSetOrder(1))%>
<%SA1->(DBGOTOP())%>

<%While !SA1->(EOF())%>

<tr>
<td width="30%">
<font size="2" face="Verdana,Arial,Helvetica,sans-serif">
<%=SA1->A1_COD%>
</font>
</td>
<td width="70%">
<font size="2" face="Verdana,Arial,Helvetica,sans-serif">
<%=SA1->A1_NOME%>
</font>
</td>
</tr>

<%SA1->(dbskip())%>

<%End%>
</table>

<p><a href="u_Inicio.apw">Voltar ao inicio</a></p>
</body>
</html>
```

## Exercício

5. Criar uma página da Web em ADVPL ASP conforme exemplo ASP5.

Com os seguintes campos para digitação do usuário:

**Nome, Email,** Endereco, Telefone, Municipio, Cidade, Cep

Estado (trazer a combo de acordo com o dicionário)

**CPF/CNPJ, Tipo** (trazer a combo de acordo com o dicionário)

- Tratar campos em azul como obrigatórios
- Ao confirmar devera se incluído o cliente utilizando MSexecauto da função MATA030.
- Utilize getsxenum para o código do cliente e retorne no browse
  - "Cliente Cadastrado com código " + **CODIGO GERADO**